
 Macro as seen in The WordPerfectionist,  July, 1992, 

“Macro of the Month: Special Character Menu”, with Alison B. Bushon

Applies to: WordPerfect/DOS, version 5.1–details, WordPerfect, all versions–character sets

I just got a scrawled note from the boss. "Type this memo for me, O.K.? I've got to run off to a

meeting. Thanx. T. B. (The Boss)" No problem, how difficult can typing a memo be? Let's see, what's

this say?

"Dear Señor Rodriguez:

Although the section of the code that you mention in your letter (§5.231) does allow outside

contractors in the bidding process, one of the paragraphs that follows (¶ 11.2) also prevents us from

allowing more than e of the total contracts for the project to be assigned to contractors from outside

the U.S. or the U.K., and we have already reached that limitation, ± £1.7 million. If any of the vital

numbers should happen to change, I will notify you j your embassy.

Sincerely, 

T. B."

All those special little symbols T. B. likes to put in his memos just can't be typed on his old

Underwood typewriter, so guess who has to take care of them? I'm just lucky that I'm got WordPerfect

5.1's Compose feature to help me. There are over 1500 characters available for T. B.'s little tricks, and

all of them can be printed on any printer that can print graphics. (WordPerfect 5.0 uses the same

character sets, but only the characters supported by your printer are available.) Whenever

WordPerfect 5.1 needs to print one of these characters, it checks your printer driver to see if that

character is supported and can be printed normally. If the character is not directly available, WP will

print the character as a small graphic, using information from the WP.DRS file, or Driver Resource

file.

Every special character has a character number, and similar characters are arranged in sets. Both

the set number and the character number are needed to use a character with the Compose feature.

These numbers are available in several different places. There is a chart in Appendix P of the

WordPerfect 5.1 manual, and the same diagram is in the file CHARMAP.TST, distributed with

WordPerfect. (Both of these charts show the special characters arranged in columns. To calculate any

character's number, read across the top line of the chart to the number directly over the desired

character. This number will be in the range from zero to 29. Add this number to the value at the left

end of the row showing the character.)

This prints all characters in Character Map 1 (Multinational)

0 1 2

0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 

000 ! " # $ % & ' ( ) * + , - . / 0 1 2 3 4 5 6 7 ß 9 : Á á Â â

030 Ä ä À à Å å Æ æ Ç ç É é Ê ê Ë ë È è Í í Î î Ï ï Ì ì Ñ ñ Ó ó

060 Ô ô Ö ö Ò ò Ú ú Û û Ü ü Ù ù Ÿ ÿ Ã ã o p Ø ø Õ õ Ý ý Ð ð Þ þ

090 { | } ~   ¡ ¢ £ ¤ ¥ ¦ § ¨ © ª « ¬ ­ ® ¯ ° ± ² ³ ´ µ ¶ · ¸ ¹
120 º » ¼ ½ ¾ ¿ À Á Â Ã Ä Å Æ Ç È É Ê Ë Ì Í Î Ï Ð Ñ Ò Ó Ô Õ Ö ×
150 Ø Ù Ú Û Ü Ý Þ ß à á â ã ä å æ ç Œ œ ê ë ì í î ï ð ñ Š š ô õ
180 ö ÷ ø ù ú û ü ý þ ÿ Ā ā Ă ă Ą ą Ć ć Ĉ ĉ Ċ ċ Č č Ď ď � � Ē ē
210 Ĕ ĕ Ė ė Ę ę Ě ě Ĝ ĝ Ğ ğ ! " # $ % & ' ( ) * + , - . / 0 1 2
240 3 4

0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9 0 1 2 3 4 5 6 7 8 9

0 1 2

There is a longer listing of all these characters in the file CHARACTR.DOC, also distributed with

WordPerfect. This file lists both the characters themselves, and the names of the symbols and letters.

Because not every character will display on screen, this file will be easier to use than CHARMAP.TST

for looking up a character without printing out the file.

Using Compose is easy, once you know the character set and number of the character you need. For

example, a bullet is available in set number four, as character number zero. To add a bullet to your

text, just press [Compose/Ctrl-V], and at the "Key =" prompt, type "4,0" and press [Enter]. Or, choose

[Compose/Ctrl-2], type "4,0" and press [Enter]. Use the two on the top line of the keyboard, not on the

numeric keypad. This second Compose method does not display the "Key =" prompt on screen, so you

may use it in the Macro Editor or the Equation Editor, where the prompt cannot be displayed.

The character will appear on screen if your monitor/video card combination can display it;

otherwise, the character will display as a small block. Whether the character shows as a block or as a

proper character, it will print correctly, and it will display properly in Print Preview, available by

pressing [Print/Shift-F7], [6-View Document]. On VGA and EGA monitors, more characters can be

shown on the editing screen by choosing to display 512 characters and giving up the ability to see a

second typeface on screen, such as italics. To see these extra characters, press [Setup/Shift-F1], [2-

Display], [1-Colors/Fonts/Attributes], [5-512 Characters], [Exit/F7].

Finally, there's printing. Special characters are both text and graphics at the same time, so they

will print at the quality setting used for graphics on the print menu, unless text quality is set to high.

For the best results when printing text including these characters, set the text quality by pressing

[Print/Shift-F7], [T-Text Quality], [4-High], before printing.

That's easy enough. Let's look at T. B.'s note. For the "Dear Señor" opening of T. B.'s memo, I'll type

"Dear Se", then find ñ in the chart of characters—it's character number 157 in set one. Next, I press

[Compose/Ctrl-V], type "1,157", press [Enter], and type the rest of the line. For the rest of the note, I

could just look up each character as I need it, and enter the numbers with Compose, or I could spend

some time typing in a macro that will display all my favorite characters, and all of T. B.'s favorites, on

screen. SPECIAL.WPM will do just that. It displays a box on screen, containing the 45 special

typographical characters and fractions that I use most frequently. SPECIAL.WPM does not include

any foreign characters, other than a few Greek letters, but you can customize it to display the

characters that you use most often.

To enter the typographical character macro, press [Home], [Macro Define/Ctrl-F10], type "special",

and press [Enter]. For the macro description, type "fractions and typographical symbols" and press

[Enter]. If you would prefer to start the macro more quickly, you may name it with an alternate-letter

key instead of the full name by pressing [Macro Define/Ctrl-F10], and holding down [Alt] while typing

a letter that is not already in use as a macro name. If an error message such as "ALTA.WPM Already

Exists" appears, press [Cancel/F1], and try again with a different letter choice.

When the macro editing screen appears, delete the {DISPLAY OFF} code and then enter the codes

in figure 2. Enter the commands {Enter}, {^V}, and {^P} by pressing [Compose/Ctrl-V], and then [Enter],

[Ctrl-V] or [Ctrl-P]. Enter the {Screen} command by pressing [Screen/Ctrl-F3]. The remaining

commands should be chosen by pressing [Macro Commands/Ctrl-PgUp], and selecting the command

from the pop-up menu. The {^P} on line 3 of the macro is a cursor positioning command. The two

characters that follow it tell WordPerfect where to draw the box. Each of the smiley face symbols are

another type of special character, different from WordPerfect's character sets. The smiley face, and

the box-drawing characters, are part of the ASCII code number set—that's the American Standard

Code for Information Interchange, also known as the IBM character set. A complete list of these codes

is in Appendix A of the WordPerfect 5.1 manual, or can be displayed by pressing [Help/F3],

[Compose/Ctrl-V]. (Press [Enter] to exit the help screen.) The smiley faces are ASCII code one, so the

cursor will be moved up to column one, row one.

To enter each smiley face, hold down the Alt key and type [1] on the numeric keypad. When you let

go of the Alt key, the smiley face will appear on the screen. Each of the box border characters are

added to the macro in the same way; just enter each number while holding the Alt key. There are

more box-drawing characters available—try using any of the IBM/ASCII characters from 169 to 223.

Upper left double corner 6 Alt-201

Upper right double corner 7 Alt-187

Horizontal double line 4 Alt-205

Vertical double line 5 Alt-186

Lower left double corner 9 Alt-200

Lower right double corner 8 Alt-188

Vertical single line * Alt-179

The display box in SPECIAL.WPM is nearly the full width of the screen, so it is slightly larger

than the Macro Editor screen, and each line of the box display will wrap to the next line. Count your

spaces carefully while entering these lines, but if the columns don't line up perfectly when you run the

macro, you can correct them easily; just press [Macro Define/Ctrl-F10], type "special" again, choose [2-

Edit], make any corrections needed, and press [Exit/F7] when done.

To run the macro, press [Macro/Alt-F10], type "special", and press [Enter]. The display box will

appear on screen. Press the letter, numbers, or symbol next to the character you would like to add to

your text. If you need to exit the macro without choosing a character, press the space bar, [Enter], or

[Cancel/F1].

Well, now you have the typographical symbols most used by T. B.'s little memos, all packed up for

easy access in a macro. But how about your boss's favorites? Easy enough. First, use [List/F5] to

display your macro directory. Place the cursor on "special.wpm" and press [8-Copy]. Enter the new

filename "yourboss.wpm", press [Enter], and choose [Exit/F7] to return to the document screen. You

may then edit the new macro normally, and create a new box full of characters that will impress Y. B.

as much as special.wpm has impressed T. B. 

{;}SPECIAL.WPM / J. Stern / 10/91~
{ON CANCEL}{GO}done~~
{CHAR}n~{^P}^A^A
  6444444444444444444444444444444444444444444444444444444444444444444444447{Enter}
   5    Choose a special character, or press the space bar to continue.    5{Enter}
   5 1 1/8 * a care/of   * j box solid   * s section   * ! heart           5{Enter}
   5 2 1/4 * b cents     * k box hollow  * t trademark * ? diamond         5{Enter}
   5 3 3/8 * c copyright * l quotes left * u slashed 0 * . club            5{Enter}
   5 4 1/2 * d degree    * m em dash     * v alpha     * , spade           5{Enter}
   5 5 5/8 * e division  * n en dash     * w beta      * / plus or minus   5{Enter}
   5 6 3/4 * f dagger 1  * o 3/4 em-dash * x delta     * $ Rx prescription 5{Enter}
   5 7 7/8 * g dagger 2  * p paragraph   * y pi        * > greater or =    5{Enter}
   5 8 1/3 * h hand left * q quotes right* z omega     * < less or =       5{Enter}
   5 9 2/3 * i hand right* r registered  * * check mark* = approximately   5{Enter}
   9444444444444444444444444444444444444444444444444444444444444444444444448~
{CASE CALL}{ORIGINAL KEY}~
        1~F1~2~F2~3~F3~4~F4~5~F5~
        6~F6~7~F7~8~F8~9~F9~
        a~Fa~b~Fb~c~Fc~d~Fd~e~Fe~
        f~Ff~g~Fg~h~Fh~i~Fi~
        j~Fj~k~Fk~l~Fl~m~Fm~n~Fn~
        o~Fo~p~Fp~q~Fq~r~Fr~
        s~Fs~t~Ft~u~Fu~v~Fv~w~Fw~
        x~Fx~y~Fy~z~Fz~*~F*~
        !~F!~?~F?~.~F.~,~F,~/~F/~
        $~F$~>~F>~<~F<~=~F=~
~
{GO}done~
{LABEL}F1~{^V}4,66{Enter}{RETURN}
{LABEL}F2~{^V}4,18{Enter}{RETURN}
{LABEL}F3~{^V}4,67{Enter}{RETURN}
{LABEL}F4~{^V}4,17{Enter}{RETURN}
{LABEL}F5~{^V}4,68{Enter}{RETURN}
{LABEL}F6~{^V}4,25{Enter}{RETURN}
{LABEL}F7~{^V}4,69{Enter}{RETURN}
{LABEL}F8~{^V}4,64{Enter}{RETURN}
{LABEL}F9~{^V}4,65{Enter}{RETURN}
{LABEL}Fa~{^V}4,73{Enter}{RETURN}
{LABEL}Fb~{^V}4,19{Enter}{RETURN}
{LABEL}Fc~{^V}4,23{Enter}{RETURN}
{LABEL}Fd~{^V}6,36{Enter}{RETURN}
{LABEL}Fe~{^V}6,8{Enter}{RETURN}
{LABEL}Ff~{^V}4,39{Enter}{RETURN}
{LABEL}Fg~{^V}4,40{Enter}{RETURN}
{LABEL}Fh~{^V}5,21{Enter}{RETURN}
{LABEL}Fi~{^V}5,22{Enter}{RETURN}
{LABEL}Fj~{^V}4,46{Enter}{RETURN}
{LABEL}Fk~{^V}4,48{Enter}{RETURN}
{LABEL}Fl~{^V}4,30{Enter}{RETURN}
{LABEL}Fm~{^V}4,34{Enter}{RETURN}
{LABEL}Fn~{^V}4,33{Enter}{RETURN}
{LABEL}Fo~{^V}4,77{Enter}{RETURN}
{LABEL}Fp~{^V}4,5{Enter}{RETURN}
{LABEL}Fq~{^V}4,31{Enter}{RETURN}
{LABEL}Fr~{^V}4,22{Enter}{RETURN}

{LABEL}Fs~{^V}4,6{Enter}{RETURN}
{LABEL}Ft~{^V}4,41{Enter}{RETURN}
{LABEL}Fu~{^V}1,80{Enter}{RETURN}
{LABEL}Fv~{^V}8,1{Enter}{RETURN}
{LABEL}Fw~{^V}8,3{Enter}{RETURN}
{LABEL}Fx~{^V}8,8{Enter}{RETURN}
{LABEL}Fy~{^V}8,33{Enter}{RETURN}
{LABEL}Fz~{^V}8,51{Enter}{RETURN}
{LABEL}F*~{^V}5,23{Enter}{RETURN}
{LABEL}F!~{^V}5,0{Enter}{RETURN}
{LABEL}F?~{^V}5,1{Enter}{RETURN}
{LABEL}F.~{^V}5,2{Enter}{RETURN}
{LABEL}F,~{^V}5,3{Enter}{RETURN}
{LABEL}F/~{^V}6,1{Enter}{RETURN}
{LABEL}F$~{^V}4,43{Enter}{RETURN}
{LABEL}F>~{^V}6,3{Enter}{RETURN}
{LABEL}F<~{^V}6,2{Enter}{RETURN}
{LABEL}F=~{^V}6,14{Enter}{RETURN}
{LABEL}done~
{Screen}3

{;}FOREIGN.WPM / J. Stern / 10/91~
{ON CANCEL}{GO}done~~
{CHAR}n~{^P}^A^A
  644444444444444444444444444444444444444444444444444444444444444444444447{Enter}
   5Choose a regular or capital foreign character, or press space to exit.5{Enter}
   5 a a-accent     * j i-circumflex * s double s    * ! invert !         5{Enter}
   5 b a-circumflex * k i-grave      * t thorn       * ? invert ?         5{Enter}
   5 c a-umlaut     * l i-umlaut     * u u-accent    * . Yen              5{Enter}
   5 d ae digraph   * m eth          * v u-circumflex* , Pesetas          5{Enter}
   5 e e-accent     * n n-tilde      * w u-grave     * / Lire             5{Enter}
   5 f e-circumflex * o o-accent     * x u-tilde     * $ Pounds U.K.      5{Enter}
   5 g e-grave      * p o-circumflex * y u-umlaut    * > Florins          5{Enter}
   5 h e-umlaut     * q o-grave      * z oe digraph  * < Francs           5{Enter}
   5 i i-accent     * r o-umlaut     * 0 slashed 0   * = European currency5{Enter}
   944444444444444444444444444444444444444444444444444444444444444444444448~
{CASE CALL}{VARIABLE}n~~
        a~Fa~b~Fb~c~Fc~d~Fd~e~Fe~
        f~Ff~g~Fg~h~Fh~i~Fi~
        j~Fj~k~Fk~l~Fl~m~Fm~n~Fn~
        o~Fo~p~Fp~q~Fq~r~Fr~
        s~Fs~t~Ft~u~Fu~v~Fv~w~Fw~
        x~Fx~y~Fy~z~Fz~
        A~GA~B~GB~C~GC~D~GD~E~GE~
        F~GF~G~GG~H~GH~I~GI~
        J~GJ~K~GK~L~GL~M~GM~N~GN~
        O~GO~P~GP~Q~GQ~R~GR~
        S~GS~T~GT~U~GU~V~GV~W~GW~
        X~GX~Y~GY~Z~GZ~0~F0~
        !~F!~?~F?~.~F.~,~F,~/~F/~
        $~F$~>~F>~<~F<~=~F=~
~
{GO}done~
{LABEL}Fa~{^V}1,27{Enter}{RETURN}
{LABEL}Fb~{^V}1,29{Enter}{RETURN}
{LABEL}Fc~{^V}1,31{Enter}{RETURN}
{LABEL}Fd~{^V}1,37{Enter}{RETURN}
{LABEL}Fe~{^V}1,41{Enter}{RETURN}
{LABEL}Ff~{^V}1,43{Enter}{RETURN}
{LABEL}Fg~{^V}1,47{Enter}{RETURN}
{LABEL}Fh~{^V}1,45{Enter}{RETURN}
{LABEL}Fi~{^V}1,49{Enter}{RETURN}
{LABEL}Fj~{^V}1,51{Enter}{RETURN}
{LABEL}Fk~{^V}1,55{Enter}{RETURN}
{LABEL}Fl~{^V}1,53{Enter}{RETURN}
{LABEL}Fm~{^V}1,87{Enter}{RETURN}
{LABEL}Fn~{^V}1,57{Enter}{RETURN}
{LABEL}Fo~{^V}1,59{Enter}{RETURN}
{LABEL}Fp~{^V}1,61{Enter}{RETURN}
{LABEL}Fq~{^V}1,65{Enter}{RETURN}
{LABEL}Fr~{^V}1,63{Enter}{RETURN}
{LABEL}Fs~{^V}1,23{Enter}{RETURN}
{LABEL}Ft~{^V}1,89{Enter}{RETURN}
{LABEL}Fu~{^V}1,67{Enter}{RETURN}
{LABEL}Fv~{^V}1,69{Enter}{RETURN}
{LABEL}Fw~{^V}1,73{Enter}{RETURN}

{LABEL}Fx~{^V}1,199{Enter}{RETURN}
{LABEL}Fy~{^V}1,71{Enter}{RETURN}
{LABEL}Fz~{^V}1,167{Enter}{RETURN}
{LABEL}GA~{^V}1,26{Enter}{RETURN}
{LABEL}GB~{^V}1,28{Enter}{RETURN}
{LABEL}GC~{^V}1,30{Enter}{RETURN}
{LABEL}GD~{^V}1,36{Enter}{RETURN}
{LABEL}GE~{^V}1,40{Enter}{RETURN}
{LABEL}GF~{^V}1,42{Enter}{RETURN}
{LABEL}GG~{^V}1,46{Enter}{RETURN}
{LABEL}GH~{^V}1,44{Enter}{RETURN}
{LABEL}GI~{^V}1,48{Enter}{RETURN}
{LABEL}GJ~{^V}1,50{Enter}{RETURN}
{LABEL}GK~{^V}1,54{Enter}{RETURN}
{LABEL}GL~{^V}1,52{Enter}{RETURN}
{LABEL}GM~{^V}1,86{Enter}{RETURN}
{LABEL}GN~{^V}1,56{Enter}{RETURN}
{LABEL}GO~{^V}1,58{Enter}{RETURN}
{LABEL}GP~{^V}1,60{Enter}{RETURN}
{LABEL}GQ~{^V}1,64{Enter}{RETURN}
{LABEL}GR~{^V}1,62{Enter}{RETURN}
{LABEL}GS~{^V}1,23{Enter}{RETURN}
{LABEL}GT~{^V}1,88{Enter}{RETURN}
{LABEL}GU~{^V}1,66{Enter}{RETURN}
{LABEL}GV~{^V}1,68{Enter}{RETURN}
{LABEL}GW~{^V}1,72{Enter}{RETURN}
{LABEL}GX~{^V}1,198{Enter}{RETURN}
{LABEL}GY~{^V}1,70{Enter}{RETURN}
{LABEL}GZ~{^V}1,166{Enter}{RETURN}
{LABEL}F0~{^V}1,80{Enter}{RETURN}
{LABEL}F!~{^V}4,7{Enter}{RETURN}
{LABEL}F?~{^V}4,8{Enter}{RETURN}
{LABEL}F.~{^V}4,12{Enter}{RETURN}
{LABEL}F,~{^V}4,13{Enter}{RETURN}
{LABEL}F/~{^V}4,61{Enter}{RETURN}
{LABEL}F$~{^V}4,11{Enter}{RETURN}
{LABEL}F>~{^V}4,14{Enter}{RETURN}
{LABEL}F<~{^V}4,58{Enter}{RETURN}
{LABEL}F=~{^V}4,60{Enter}{RETURN}
{LABEL}done~
{Screen}3

